

---

Subject: Re: Log-scaled colorbar example  
Posted by [David Fanning](#) on Mon, 23 May 2011 15:37:38 GMT  
[View Forum Message](#) <> [Reply to Message](#)

---

Kim writes:

> After several unsuccessful attempts at trying to generate a log-scaled  
> colorbar using the COLORBAR function, I came up with a way to manually  
> create the colorbar. If someone could please look at the code and let  
> me know if this seems correct I would really appreciate it.

Well, your plots are beautiful, but I am not sure they  
are correct. :-)

Here is code that I \*know\* produces the correct result:

```
image = cgDemoData(7) ; World Elevation Data
image = Scale_Vector(image, 1, 2500L)
logImage = Alog10(image)
s = Size(image, /Dimensions)
cgDisplay, 600, 400, Title='Logarithmic Color Bar'
LoadCT, 33, /Silent, NColors=254
pos = [0.1, 0.1, 0.8, 0.9]
cglImage, BytScl(logImage, Top=253), Position=pos, /Keep_Aspect
cgPlot, [0], [0], /NoErase, XRange=[0,s[0]], YRange=[0,s[1]], $
    XStyle=1, YStyle=1, Color=cgColor('black', 255), Position=pos
cgColorbar, /YLOG, YTICKS=0, Range=[Min(image), Max(image)], $
    NColors=254, Color=cgColor('black', 255), /Vertical
END
```

How do I know? I can check it.

```
IDL> cglImageInfo, image
```

You can click in the graphics window and the image position  
and value will be printed in the command log. (Right click  
to exit the cglImageInfo program.) You can clearly see that  
the image value corresponds to the colors on the color bar.

Can you reproduce this figure and color bar with your  
code? (I haven't been able to, but I've spent less than  
an hour trying.) When you can reproduce this figure, then  
I think you are on the right track.

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.idlcoyote.com/>

Sepore ma de ni thui. ("Perhaps thou speakest truth.")

---